

Exercitium (curso 2018–19)  
Ejercicios de programación funcional con Haskell  
(hasta el 7 de diciembre de 2018)

José A. Alonso Jiménez

---

Grupo de Lógica Computacional  
Dpto. de Ciencias de la Computación e Inteligencia Artificial  
Universidad de Sevilla  
Sevilla, 16 de diciembre de 2018

Esta obra está bajo una licencia Reconocimiento–NoComercial–CompartirIgual 2.5 Spain de Creative Commons.

**Se permite:**

- copiar, distribuir y comunicar públicamente la obra
- hacer obras derivadas

**Bajo las condiciones siguientes:**

**Reconocimiento.** Debe reconocer los créditos de la obra de la manera especificada por el autor.



**No comercial.** No puede utilizar esta obra para fines comerciales.



**Compartir bajo la misma licencia.** Si altera o transforma esta obra, o genera una obra derivada, sólo puede distribuir la obra generada bajo una licencia idéntica a ésta.

- Al reutilizar o distribuir la obra, tiene que dejar bien claro los términos de la licencia de esta obra.
- alguna de estas condiciones puede no aplicarse si se obtiene el permiso del titular de los derechos de autor.

Esto es un resumen del texto legal (la licencia completa). Para ver una copia de esta licencia, visite <http://creativecommons.org/licenses/by-nc-sa/2.5/es/> o envíe una carta a Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

*Para Guiomar*



# Índice general

1	Listas equidigitales	9
2	Distancia de Hamming	11
3	Último dígito no nulo del factorial	15
4	Diferencia simétrica	19
5	Números libres de cuadrados	21
6	Capicúas productos de dos números de dos dígitos	25
7	Números autodescriptivos	27
8	Número de parejas	29
9	Reconocimiento de particiones	33
10	Relación definida por una partición	37
11	Ceros finales del factorial	39
12	Números primos sumas de dos primos	43
13	Suma de inversos de potencias de cuatro	47
14	Elemento solitario	51
15	Números colinas	57
16	Raíz cúbica entera	61
17	Numeración de los árboles binarios completos	65

<b>18 Posiciones en árboles binarios</b>	<b>67</b>
<b>19 Posiciones en árboles binarios completos</b>	<b>71</b>

# Introducción

*"The chief goal of my work as an educator and author is to help people learn to write beautiful programs."*

(Donald Knuth en [Computer programming as an art](#))

Este libro es una recopilación de las soluciones de los ejercicios propuestos en el blog [Exercitium](#) <sup>1</sup> durante el curso 2018-19.

El principal objetivo de Exercitium es servir de complemento a la asignatura de [Informática](#) <sup>2</sup> de 1º del Grado en Matemáticas de la Universidad de Sevilla.

Con los problemas de Exercitium, a diferencia de los de las [relaciones](#) <sup>3</sup>, se pretende practicar con los conocimientos adquiridos durante todo el curso, mientras que con las relaciones están orientadas a los nuevos conocimientos.

Habitualmente de cada ejercicio se muestra distintas soluciones y se compara sus eficiencias.

La dinámica del blog es la siguiente: cada día, de lunes a viernes, se propone un ejercicio para que los alumnos escriban distintas soluciones en los comentarios. Pasado 7 días de la propuesta de cada ejercicio, se cierra los comentarios y se publica una selección de sus soluciones.

Para conocer la cronología de los temas explicados se puede consultar el [diario de clase](#) <sup>4</sup>.

En el libro se irán añadiendo semanalmente las soluciones de los ejercicios del curso.

---

<sup>1</sup><https://www.glc.us.es/~jalonso/exercitium>

<sup>2</sup><https://www.cs.us.es/~jalonso/cursos/ilm-18>

<sup>3</sup><https://www.cs.us.es/~jalonso/cursos/ilm-18/ejercicios/ejercicios-I1M-2018.pdf>

<sup>4</sup><https://www.glc.us.es/~jalonso/vestigium/category/curso/ilm/ilm2018>

El código del libro se encuentra en [GitHub](#) <sup>5</sup>

---

<sup>5</sup><https://github.com/jaalonso/Exercitium2018>



# Ejercicio 1

## Listas equidigitales

### Ejercicio propuesto el 9-11-18

Una lista de números naturales es equidigital si todos sus elementos tienen el mismo número de dígitos.

Definir la función

---

```
equidigital :: [Int] -> Bool
```

---

tal que (equidigital xs) se verifica si xs es una lista equidigital. Por ejemplo,

---

```
equidigital [343,225,777,943] == True
equidigital [343,225,777,94,3] == False
```

---

## Soluciones

```
-- 1ª definición
-- =====
```

```
equidigital :: [Int] -> Bool
equidigital xs = todosIguales (numerosDeDigitos xs)
```

```
-- (numerosDeDigitos xs) es la lista de los números de dígitos de
```

```

-- los elementos de xs. Por ejemplo,
--   numerosDeDigitos [343,225,777,943] == [3,3,3,3]
--   numerosDeDigitos [343,225,777,94,3] == [3,3,3,2,1]
numerosDeDigitos :: [Int] -> [Int]
numerosDeDigitos xs = [numeroDeDigitos x | x <- xs]

-- (numeroDeDigitos x) es el número de dígitos de x. Por ejemplo,
--   numeroDeDigitos 475 == 3
numeroDeDigitos :: Int -> Int
numeroDeDigitos x = length (show x)

-- (todosIguales xs) se verifica si todos los elementos de xs son
-- iguales. Por ejemplo,
--   todosIguales [3,3,3,3] == True
--   todosIguales [3,3,3,2,1] == False
todosIguales :: Eq a => [a] -> Bool
todosIguales (x:y:zs) = x == y && todosIguales (y:zs)
todosIguales _        = True

-- 2ª definición
-- =====

equidigital2 :: [Int] -> Bool
equidigital2 []      = True
equidigital2 (x:xs) = and [numeroDeDigitos y == n | y <- xs]
                    where n = numeroDeDigitos x

-- 3ª definición
-- =====

equidigital3 :: [Int] -> Bool
equidigital3 (x:y:zs) = numeroDeDigitos x == numeroDeDigitos y &&
                        equidigital3 (y:zs)
equidigital3 _        = True

```

## Ejercicio 2

# Distancia de Hamming

### Ejercicio propuesto el 10-11-18

La distancia de Hamming entre dos listas es el número de posiciones en que los correspondientes elementos son distintos. Por ejemplo, la distancia de Hamming entre `romaz` y `loba` es 2 (porque hay 2 posiciones en las que los elementos correspondientes son distintos: la 1ª y la 3ª).

Definir la función

---

```
distancia :: Eq a => [a] -> [a] -> Int
```

---

tal que `(distancia xs ys)` es la distancia de Hamming entre `xs` e `ys`. Por ejemplo,

---

```
distancia "romano" "comino" == 2
distancia "romano" "camino" == 3
distancia "roma"   "comino" == 2
distancia "roma"   "camino" == 3
distancia "romano" "ron"    == 1
distancia "romano" "cama"   == 2
distancia "romano" "rama"   == 1
```

---

Comprobar con QuickCheck si la distancia de Hamming tiene la siguiente propiedad: `distancia(xs,ys) = 0` si, y sólo si, `xs = ys` y, en el caso de que no se verifique, modificar ligeramente la propiedad para obtener una condición necesaria y suficiente de `distancia(xs,ys) = 0`.

## Soluciones

```
import Test.QuickCheck
```

```
-- 1ª definición:
```

```
distancia :: Eq a => [a] -> [a] -> Int
distancia xs ys = length [(x,y) | (x,y) <- zip xs ys, x /= y]
```

```
-- 2ª definición:
```

```
distancia2 :: Eq a => [a] -> [a] -> Int
distancia2 [] _ = 0
distancia2 _ [] = 0
distancia2 (x:xs) (y:ys) | x /= y = 1 + distancia2 xs ys
                        | otherwise = distancia2 xs ys
```

```
-- La propiedad es
```

```
prop_distancial :: [Int] -> [Int] -> Bool
prop_distancial xs ys =
  (distancia xs ys == 0) == (xs == ys)
```

```
-- La comprobación es
```

```
-- ghci> quickCheck prop_distancial
-- *** Failed! Falsifiable (after 2 tests and 1 shrink):
-- []
-- [1]
```

```
-- En efecto,
```

```
-- ghci> distancia [] [1] == 0
-- True
-- ghci> [] == [1]
-- False
```

```
-- La primera modificación es restringir la propiedad a lista de igual
-- longitud:
```

```
prop_distancia2 :: [Int] -> [Int] -> Property
prop_distancia2 xs ys =
  length xs == length ys ==>
  (distancia xs ys == 0) == (xs == ys)
```

```
-- La comprobación es
```

```
-- ghci> quickCheck prop_distancia2
-- *** Gave up! Passed only 33 tests.

-- Nota. La propiedad se verifica, pero al ser la condición demasiado
-- restringida sólo 33 de los casos la cumple.

-- La segunda restricción es limitar las listas a la longitud de la más
-- corta:
prop_distancia3 :: [Int] -> [Int] -> Bool
prop_distancia3 xs ys =
  (distancia xs ys == 0) == (take n xs == take n ys)
  where n = min (length xs) (length ys)

-- La comprobación es
-- ghci> quickCheck prop_distancia3
-- +++ OK, passed 100 tests.
```



## Ejercicio 3

# Último dígito no nulo del factorial

### Ejercicio propuesto el 13-11-18

El factorial de 7 es  $7! = 1 * 2 * 3 * 4 * 5 * 6 * 7 = 5040$ . Por tanto, el último dígito no nulo del factorial de 7 es 4.

Definir la función

---

```
ultimoNoNuloFactorial :: Integer -> Integer
```

---

tal que `(ultimoNoNuloFactorial n)` es el último dígito no nulo del factorial de `n`. Por ejemplo,

---

```
ultimoNoNuloFactorial 7  == 4
ultimoNoNuloFactorial 10 == 8
ultimoNoNuloFactorial 12 == 6
ultimoNoNuloFactorial 97 == 2
ultimoNoNuloFactorial 0  == 1
```

---

Comprobar con QuickCheck que si `n` es mayor que 4, entonces el último dígito no nulo del factorial de `n` es par.

## Solución

```

import Test.QuickCheck

-- 1ª definición
-- =====

ultimoNoNuloFactorial :: Integer -> Integer
ultimoNoNuloFactorial n = ultimoNoNulo (factorial n)

-- (ultimoNoNulo n) es el último dígito no nulo de n. Por ejemplo,
--     ultimoNoNulo 5040 == 4
ultimoNoNulo :: Integer -> Integer
ultimoNoNulo n
  | m /= 0    = m
  | otherwise = ultimoNoNulo (n `div` 10)
  where m = n `rem` 10

-- (factorial n) es el factorial de n. Por ejemplo,
--     factorial 7 == 5040
factorial :: Integer -> Integer
factorial n = product [1..n]

-- 2ª definición
-- =====

ultimoNoNuloFactorial2 :: Integer -> Integer
ultimoNoNuloFactorial2 n = ultimoNoNulo2 (factorial n)

-- (ultimoNoNulo2 n) es el último dígito no nulo de n. Por ejemplo,
--     ultimoNoNulo 5040 == 4
ultimoNoNulo2 :: Integer -> Integer
ultimoNoNulo2 n = read [head (dropWhile (=='0') (reverse (show n)))]

-- Comprobación
-- =====

-- La propiedad es
prop_ultimoNoNuloFactorial :: Integer -> Property
prop_ultimoNoNuloFactorial n =

```



---

```
n > 4 ==> even (ultimoNoNuloFactorial n)

-- La comprobación es
--   ghci> quickCheck prop_ultimoNoNuloFactorial
--   +++ OK, passed 100 tests.
```



## Ejercicio 4

### Diferencia simétrica

#### Ejercicio propuesto el 14-11-18

La **diferencia simétrica** de dos conjuntos es el conjunto cuyos elementos son aquellos que pertenecen a alguno de los conjuntos iniciales, sin pertenecer a ambos a la vez. Por ejemplo, la diferencia simétrica de 2,5,3 y 4,2,3,7 es 5,4,7.

Definir la función

---

```
diferenciaSimetrica :: Ord a => [a] -> [a] -> [a]
```

---

tal que (diferenciaSimetrica xs ys) es la diferencia simétrica de xs e ys. Por ejemplo,

---

```
diferenciaSimetrica [2,5,3] [4,2,3,7] == [4,5,7]
diferenciaSimetrica [2,5,3] [5,2,3]   == []
diferenciaSimetrica [2,5,2] [4,2,3,7] == [3,4,5,7]
diferenciaSimetrica [2,5,2] [4,2,4,7] == [4,5,7]
diferenciaSimetrica [2,5,2,4] [4,2,4,7] == [5,7]
```

---

## Soluciones

```
import Data.List
```

```
-- 1ª definición
```

```
diferenciaSimetrica :: Ord a => [a] -> [a] -> [a]
diferenciaSimetrica xs ys =
  sort (nub ([x | x <- xs, x 'notElem' ys] ++ [y | y <- ys, y 'notElem' xs]))

-- 2ª definición
diferenciaSimetrica2 :: Ord a => [a] -> [a] -> [a]
diferenciaSimetrica2 xs ys =
  sort (nub (union xs ys \\ intersect xs ys))

-- 3ª definición
diferenciaSimetrica3 :: Ord a => [a] -> [a] -> [a]
diferenciaSimetrica3 xs ys =
  [x | x <- sort (nub (xs ++ ys))
    , x 'notElem' xs || x 'notElem' ys]
```

## Ejercicio 5

# Números libres de cuadrados

### Ejercicio propuesto el 15-11-18

Un número entero positivo es libre de cuadrados si no es divisible por el cuadrado de ningún entero mayor que 1. Por ejemplo, 70 es libre de cuadrado porque sólo es divisible por 1, 2, 5, 7 y 70; en cambio, 40 no es libre de cuadrados porque es divisible por  $2^2$ .

Definir la función

---

```
libreDeCuadrados :: Integer -> Bool
```

---

tal que (libreDeCuadrados x) se verifica si x es libre de cuadrados. Por ejemplo,

---

libreDeCuadrados 70	==	True
libreDeCuadrados 40	==	False
libreDeCuadrados 510510	==	True
libreDeCuadrados (((10 <sup>10</sup> ) <sup>10</sup> ) <sup>10</sup> )	==	False

---

## Soluciones

```
import Data.List (nub)
```

```
-- 1ª definición
```

```

-- =====

libreDeCuadrados :: Integer -> Bool
libreDeCuadrados x = x == product (divisoresPrimos x)

-- (divisoresPrimos x) es la lista de los divisores primos de x. Por
-- ejemplo,
--     divisoresPrimos 40 == [2,5]
--     divisoresPrimos 70 == [2,5,7]
divisoresPrimos :: Integer -> [Integer]
divisoresPrimos x = [n | n <- divisores x, primo n]

-- (divisores n) es la lista de los divisores del número n. Por ejemplo,
--     divisores 30 == [1,2,3,5,6,10,15,30]
divisores :: Integer -> [Integer]
divisores n = [x | x <- [1..n], n `mod` x == 0]

-- (primo n) se verifica si n es primo. Por ejemplo,
--     primo 30 == False
--     primo 31 == True
primo :: Integer -> Bool
primo n = divisores n == [1, n]

-- 2ª definición
-- =====

libreDeCuadrados2 :: Integer -> Bool
libreDeCuadrados2 n =
    null [x | x <- [2..n], rem n (x^2) == 0]

-- 3ª definición
-- =====

libreDeCuadrados3 :: Integer -> Bool
libreDeCuadrados3 n =
    null [x | x <- [2..floor (sqrt (fromIntegral n))],
            rem n (x^2) == 0]

-- 4ª definición
-- =====

```

```
libreDeCuadrados4 :: Integer -> Bool
libreDeCuadrados4 x =
    factorizacion x == nub (factorizacion x)

-- (factorizacion n) es la lista de factores primos de n. Por ejemplo,
--   factorizacion 180 == [2,2,3,3,5]
factorizacion :: Integer -> [Integer]
factorizacion n | n == 1    = []
                | otherwise = x : factorizacion (div n x)
    where x = menorFactor n

-- (menorFactor n) es el menor divisor de n. Por ejemplo,
--   menorFactor 15 == 3
menorFactor :: Integer -> Integer
menorFactor n = head [x | x <- [2..], rem n x == 0]

-- Comparación de eficiencia
-- =====

--   λ> libreDeCuadrados 510510
--   True
--   (0.76 secs, 89,522,360 bytes)
--   λ> libreDeCuadrados2 510510
--   True
--   (1.78 secs, 371,826,320 bytes)
--   λ> libreDeCuadrados3 510510
--   True
--   (0.01 secs, 0 bytes)
--   λ> libreDeCuadrados4 510510
--   True
--   (0.00 secs, 153,216 bytes)
```





## Ejercicio 6

# Capicúas productos de dos números de dos dígitos

### Ejercicio propuesto el 16 de noviembre de 2018

El número 9009 es capicúa y es producto de dos números de dos dígitos, pues  $9009 = 91 \times 99$ .

Definir la lista

---

```
capicuasP2N2D :: [Int]
```

---

cuyos elementos son los números capicúas que son producto de 2 números de dos dígitos. Por ejemplo,

---

```
take 5  capicuasP2N2D == [121,242,252,272,323]
length  capicuasP2N2D == 74
drop 70  capicuasP2N2D == [8008,8118,8448,9009]
```

---

## Soluciones

```
import Data.List (nub, sort)
```

```
capicuasP2N2D :: [Int]
```

```
capicuasP2N2D = [x | x <- productos, esCapicua x]
```

```
-- productos es la lista de números que son productos de 2 números de
-- dos dígitos.
productos :: [Int]
productos = sort (nub [x*y | x <- [10..99], y <- [x..99]])

-- (esCapicua x) se verifica si x es capicúa.
esCapicua :: Int -> Bool
esCapicua x = xs == reverse xs
  where xs = show x
```

## Ejercicio 7

# Números autodescriptivos

### Ejercicio propuesto el 19-11-18

Un número  $n$  es autodescriptivo cuando para cada posición  $k$  de  $n$  (empezando a contar las posiciones a partir de 0), el dígito en la posición  $k$  es igual al número de veces que ocurre  $k$  en  $n$ . Por ejemplo, 1210 es autodescriptivo porque tiene 1 dígito igual a "0", 2 dígitos iguales a "1", 1 dígito igual a "2" y ningún dígito igual a "3".

Definir la función

---

```
autodescriptivo :: Integer -> Bool
```

---

tal que (autodescriptivo  $n$ ) se verifica si  $n$  es autodescriptivo. Por ejemplo,

---

```
λ> autodescriptivo 1210
True
λ> [x | x <- [1..100000], autodescriptivo x]
[1210,2020,21200]
λ> autodescriptivo 9210000001000
True
```

---

**Nota:** Se puede usar la función [genericLength](#).

## Soluciones

```
import Data.List (genericLength)

autodescriptivo :: Integer -> Bool
autodescriptivo n = autodescriptiva (digitos n)

digitos :: Integer -> [Integer]
digitos n = [read [d] | d <- show n]

autodescriptiva :: [Integer] -> Bool
autodescriptiva ns =
  and [x == ocurrencias k ns | (k,x) <- zip [0..] ns]

ocurrencias :: Integer -> [Integer] -> Integer
ocurrencias x ys = genericLength (filter (==x) ys)
```

## Ejercicio 8

# Número de parejas

### Ejercicio propuesto el 20-11-18

Definir la función

---

```
nParejas :: Ord a => [a] -> Int
```

---

tal que (nParejas xs) es el número de parejas de elementos iguales en xs. Por ejemplo,

---

```
nParejas [1,2,2,1,1,3,5,1,2] == 3
nParejas [1,2,1,2,1,3,2]     == 2
nParejas [1..2*10^6]         == 0
nParejas2 ([1..10^6] ++ [1..10^6]) == 1000000
```

---

En el primer ejemplos las parejas son (1,1), (1,1) y (2,2). En el segundo ejemplo, las parejas son (1,1) y (2,2).

Comprobar con QuickCheck que para toda lista de enteros xs, el número de parejas de xs es igual que el número de parejas de la inversa de xs.

## Soluciones

```
import Test.QuickCheck
import Data.List ((\\), group, sort)
```

```

-- 1ª solución
nParejas :: Ord a => [a] -> Int
nParejas [] = 0
nParejas (x:xs) | x `elem` xs = 1 + nParejas (xs \\ [x])
                | otherwise   = nParejas xs

-- 2ª solución
nParejas2 :: Ord a => [a] -> Int
nParejas2 xs =
  sum [length ys `div` 2 | ys <- group (sort xs)]

-- 3ª solución
nParejas3 :: Ord a => [a] -> Int
nParejas3 = sum . map (('div' 2) . map length . group . sort

-- 4ª solución
nParejas4 :: Ord a => [a] -> Int
nParejas4 = sum . map (('div' 2) . length) . group . sort

-- Equivalencia
prop_equiv :: [Int] -> Bool
prop_equiv xs =
  nParejas xs == nParejas2 xs &&
  nParejas xs == nParejas3 xs &&
  nParejas xs == nParejas4 xs

-- Comprobación:
--   λ> quickCheck prop_equiv
--   +++ OK, passed 100 tests.

-- Comparación de eficiencia
--   λ> nParejas [1..20000]
--   0
--   (2.54 secs, 4,442,808 bytes)
--   λ> nParejas2 [1..20000]
--   0
--   (0.03 secs, 12,942,232 bytes)
--   λ> nParejas3 [1..20000]
--   0

```

```
-- (0.02 secs, 13,099,904 bytes)
-- λ> nParejas4 [1..20000]
-- 0
-- (0.01 secs, 11,951,992 bytes)

-- Propiedad:
prop_nParejas :: [Int] -> Bool
prop_nParejas xs =
  nParejas xs == nParejas (reverse xs)

-- Comprobación
comprueba :: IO ()
comprueba = quickCheck prop_nParejas

-- Comprobación:
-- λ> comprueba
-- +++ OK, passed 100 tests.
```





## Ejercicio 9

# Reconocimiento de particiones

### Ejercicio propuesto el 21-11-18

Una **partición** de un conjunto es una división del mismo en subconjuntos disjuntos no vacíos.

Definir la función

---

```
esParticion :: Eq a => [[a]] -> Bool
```

---

tal que (esParticion xss) se verifica si xss es una partición; es decir sus elementos son listas no vacías disjuntas. Por ejemplo.

---

```
esParticion [[1,3],[2],[9,5,7]] == True
esParticion [[1,3],[2],[9,5,1]] == False
esParticion [[1,3],[],[9,5,7]]  == False
esParticion [[2,3,2],[4]]       == True
```

---

## Soluciones

```
import Data.List ((\\), intersect)
```

```
-- 1ª definición
-- =====
```

```

esParticion :: Eq a => [[a]] -> Bool
esParticion xss =
  [] 'notElem' xss &&
  and [disjuntos xs ys | xs <- xss, ys <- xss \ [xs]]

disjuntos :: Eq a => [a] -> [a] -> Bool
disjuntos xs ys = null (xs 'intersect' ys)

```

```

-- 2ª definición
-- =====

```

```

esParticion2 :: Eq a => [[a]] -> Bool
esParticion2 [] = True
esParticion2 (xs:xss) =
  not (null xs) &&
  and [disjuntos xs ys | ys <- xss] &&
  esParticion2 xss

```

```

-- 3ª definición
-- =====

```

```

esParticion3 :: Eq a => [[a]] -> Bool
esParticion3 [] = True
esParticion3 (xs:xss) =
  not (null xs) &&
  all (disjuntos xs) xss &&
  esParticion3 xss

```

```

-- Equivalencia

```

```

prop_equiv :: [[Int]] -> Bool
prop_equiv xss =
  and [esParticion xss == f xss | f <- [ esParticion2
                                          , esParticion3]]

```

```

-- Comprobación

```

```

-- λ> quickCheck prop_equiv
-- +++ OK, passed 100 tests.

```

```

-- Comparación de eficiencia:

```

```

-- λ> esParticion [[x] | x <- [1..3000]]

```

```
-- True
-- (4.37 secs, 3,527,956,400 bytes)
-- λ> esParticion2 [[x] | x <- [1..3000]]
-- True
-- (1.26 secs, 1,045,792,552 bytes)
-- λ> esParticion3 [[x] | x <- [1..3000]]
-- True
-- (1.30 secs, 1,045,795,272 bytes)
-- λ> esParticion3 [[x] | x <- [1..3000]]
-- True
-- (1.30 secs, 1,045,795,272 bytes)
```



## Ejercicio 10

# Relación definida por una partición

### Ejercicio propuesto el 22-11-18

Dos elementos están [relacionados por una partición](#) xss si pertenecen al mismo elemento de xss.

Definir la función

---

```
relacionados :: Eq a => [[a]] -> a -> a -> Bool
```

---

tal que (relacionados xss y z) se verifica si los elementos y y z están relacionados por la partición xss. Por ejemplo,

---

```
relacionados [[1,3],[2],[9,5,7]] 7 9 == True
relacionados [[1,3],[2],[9,5,7]] 3 9 == False
relacionados [[1,3],[2],[9,5,7]] 4 9 == False
```

---

## Soluciones

```
-- 1ª definición
-- =====
```

```
relacionados :: Eq a => [[a]] -> a -> a -> Bool
```

```

relacionados [] _ _ = False
relacionados (xs:xss) y z
  | y 'elem' xs = z 'elem' xs
  | otherwise  = relacionados xss y z

-- 2ª definición
-- =====

relacionados2 :: Eq a => [[a]] -> a -> a -> Bool
relacionados2 xss y z =
  or [elem y xs && elem z xs | xs <- xss]

-- 3ª definición
-- =====

relacionados3 :: Eq a => [[a]] -> a -> a -> Bool
relacionados3 xss y z =
  or [[y,z] 'subconjunto' xs | xs <- xss]

-- (subconjunto xs ys) se verifica si xs es un subconjunto de ys; es
-- decir, si todos los elementos de xs pertenecen a ys. Por ejemplo,
--   subconjunto [3,2,3] [2,5,3,5] == True
--   subconjunto [3,2,3] [2,5,6,5] == False
subconjunto :: Eq a => [a] -> [a] -> Bool
subconjunto xs ys = and [elem x ys | x <- xs]

-- 4ª definición
-- =====

relacionados4 :: Eq a => [[a]] -> a -> a -> Bool
relacionados4 xss y z =
  any ([y,z] 'subconjunto') xss

```

# Ejercicio 11

## Ceros finales del factorial

### Ejercicio propuesto el 23-11-18

Definir la función

---

```
cerosDelFactorial :: Integer -> Integer
```

---

tal que (cerosDelFactorial n) es el número de ceros en que termina el factorial de n. Por ejemplo,

---

```
cerosDelFactorial 24 == 4
cerosDelFactorial 25 == 6
length (show (cerosDelFactorial (1234^5678))) == 17552
```

---

## Soluciones

```
import Data.List (genericLength)

-- 1ª definición
-- =====

cerosDelFactorial :: Integer -> Integer
cerosDelFactorial n = ceros (factorial n)

-- (factorial n) es el factorial n. Por ejemplo,
```

```

--      factorial 3 == 6
factorial :: Integer -> Integer
factorial n = product [1..n]

-- (ceros n) es el número de ceros en los que termina el número n. Por
-- ejemplo,
--      ceros 320000 == 4
ceros :: Integer -> Integer
ceros n | rem n 10 /= 0 = 0
        | otherwise   = 1 + ceros (div n 10)

-- 2ª definición
-- =====

cerosDelFactorial2 :: Integer -> Integer
cerosDelFactorial2 n = ceros2 (factorial n)

-- (ceros n) es el número de ceros en los que termina el número n. Por
-- ejemplo,
--      ceros 320000 == 4
ceros2 :: Integer -> Integer
ceros2 n = genericLength (takeWhile (=='0') (reverse (show n)))

-- 3ª definición
-- =====

cerosDelFactorial3 :: Integer -> Integer
cerosDelFactorial3 n
  | n < 5      = 0
  | otherwise = m + cerosDelFactorial3 m
  where m = n `div` 5

-- Comparación de la eficiencia
--      λ> cerosDelFactorial1 (3*10^4)
--      7498
--      (3.96 secs, 1,252,876,376 bytes)
--      λ> cerosDelFactorial2 (3*10^4)
--      7498
--      (3.07 secs, 887,706,864 bytes)
--      λ> cerosDelFactorial3 (3*10^4)

```



```
-- 7498
-- (0.03 secs, 9,198,896 bytes)
```



## Ejercicio 12

# Números primos sumas de dos primos

### Ejercicio propuesto el 26-11-18

Definir las funciones

---

```
esPrimoSumaDeDosPrimos :: Integer -> Bool
primosSumaDeDosPrimos  :: [Integer]
```

---

tales que

- 
- (esPrimoSumaDeDosPrimos x) se verifica si x es un número primo que se puede escribir como la suma de dos números primos. Por ejemplo,

---

```
esPrimoSumaDeDosPrimos 19 == True
esPrimoSumaDeDosPrimos 20 == False
esPrimoSumaDeDosPrimos 23 == False
```

---

- primosSumaDeDosPrimos es la lista de los números primos que se pueden escribir como la suma de dos números primos. Por ejemplo,

---

```
λ> take 17 primosSumaDeDosPrimos
[5,7,13,19,31,43,61,73,103,109,139,151,181,193,199,229,241]
λ> primosSumaDeDosPrimos !! (10^5)
18409543
```

---

## Soluciones

```
import Data.Numbers.Primes (isPrime, primes)
import Test.QuickCheck

-- 1ª solución
-- =====

esPrimoSumaDeDosPrimos :: Integer -> Bool
esPrimoSumaDeDosPrimos x =
    isPrime x && isPrime (x - 2)

primosSumaDeDosPrimos :: [Integer]
primosSumaDeDosPrimos =
    [x | x <- primes
      , isPrime (x - 2)]

-- 2ª solución
-- =====

primosSumaDeDosPrimos2 :: [Integer]
primosSumaDeDosPrimos2 =
    [y | (x,y) <- zip primes (tail primes)
      , y == x + 2]

esPrimoSumaDeDosPrimos2 :: Integer -> Bool
esPrimoSumaDeDosPrimos2 x =
    x == head (dropWhile (<x) primosSumaDeDosPrimos2)

-- Equivalencias
-- =====

-- Equivalencia de esPrimoSumaDeDosPrimos
prop_esPrimoSumaDeDosPrimos_equiv :: Integer -> Property
prop_esPrimoSumaDeDosPrimos_equiv x =
    x > 0 ==>
    esPrimoSumaDeDosPrimos x == esPrimoSumaDeDosPrimos2 x

-- La comprobación es
--    λ> quickCheck prop_esPrimoSumaDeDosPrimos_equiv
```

```
--      +++ OK, passed 100 tests.

-- Equivalencia de primosSumaDeDosPrimos
prop_primosSumaDeDosPrimos_equiv :: Int -> Property
prop_primosSumaDeDosPrimos_equiv n =
  n >= 0 ==>
  primosSumaDeDosPrimos !! n == primosSumaDeDosPrimos2 !! n

-- La comprobación es
--      λ> quickCheck prop_primosSumaDeDosPrimos_equiv
--      +++ OK, passed 100 tests.

-- Comparación de eficiencia
--      =====

--      λ> primosSumaDeDosPrimos !! (10^4)
--      1261081
--      (2.07 secs, 4,540,085,256 bytes)
--
-- Se recarga para evitar memorización
--      λ> primosSumaDeDosPrimos2 !! (10^4)
--      1261081
--      (0.49 secs, 910,718,408 bytes)
```



## Ejercicio 13


# Suma de inversos de potencias de cuatro

### Ejercicio propuesto el 27-11-18

Esta semana se ha publicado en [Twitter](#) una demostración visual de que

$$1/4 + 1/4^2 + 1/4^3 + \dots = 1/3$$

como se muestra en la siguiente imagen



Definir las funciones

```
sumaInversosPotenciasDeCuatro :: [Double]
aproximacion :: Double -> Int
```

---

tales que

- sumaInversosPotenciasDeCuatro es la lista de la suma de la serie de los inversos de las potencias de cuatro. Por ejemplo,

```
λ> take 6 sumaInversosPotenciasDeCuatro
[0.25,0.3125,0.328125,0.33203125,0.3330078125,0.333251953125]
```

---

- (aproximacion e) es el menor número de términos de la serie anterior que hay que sumar para que el valor absoluto de su diferencia con  $1/3$  sea menor que e. Por ejemplo,

```
aproximacion 0.001 == 4
aproximacion 1e-3 == 4
aproximacion 1e-6 == 9
aproximacion 1e-20 == 26
sumaInversosPotenciasDeCuatro !! 26 == 0.3333333333333333
```

---

## Soluciones

-- 1ª definición

```
sumaInversosPotenciasDeCuatro :: [Double]
sumaInversosPotenciasDeCuatro =
  [sum [1 / (4^k) | k <- [1..n]] | n <- [1..]]
```

-- 2ª definición

```
sumaInversosPotenciasDeCuatro2 :: [Double]
sumaInversosPotenciasDeCuatro2 =
  [1/4*((1/4)^n-1)/(1/4-1) | n <- [1..]]
```

-- 3ª definición

```
sumaInversosPotenciasDeCuatro3 :: [Double]
sumaInversosPotenciasDeCuatro3 =
  [(1 - 0.25^n)/3 | n <- [1..]]
```

-- 1ª solución



```
aproximacion :: Double -> Int
aproximacion e =
  length (takeWhile (>=e) es)
  where es = [abs (1/3 - x) | x <- sumaInversosPotenciasDeCuatro2]
```

*-- 2ª solución*

```
aproximacion2 :: Double -> Int
aproximacion2 e =
  head [n | (x,n) <- zip es [0..]
        , x < e]
  where es = [abs (1/3 - x) | x <- sumaInversosPotenciasDeCuatro2]
```



# Ejercicio 14

## Elemento solitario

### Ejercicio propuesto el 28-11-18

Definir la función

---

```
solitario :: Ord a => [a] -> a
```

---

tal que (solitario xs) es el único elemento que ocurre una vez en la lista xs (se supone que la lista xs tiene al menos 3 elementos y todos son iguales menos uno que es el solitario). Por ejemplo,

---

```
solitario [2,2,7,2] == 7
solitario [2,2,2,7] == 7
solitario [7,2,2,2] == 7
solitario (replicate (2*10^7) 1 ++ [2]) == 2
```

---

## Soluciones

```
import Test.QuickCheck
import Data.List (group, nub, sort)

-- 1ª definición
-- =====

solitario :: Ord a => [a] -> a
```

```

solitario xs =
    head [x | x <- xs
           , cuenta xs x == 1]

cuenta :: Eq a => [a] -> a -> Int
cuenta xs x = length [y | y <- xs
                          , x == y]

-- 2ª definición
-- =====

solitario2 :: Ord a => [a] -> a
solitario2 xs = head (filter (\x -> cuenta2 xs x == 1) xs)

cuenta2 :: Eq a => [a] -> a -> Int
cuenta2 xs x = length (filter (==x) xs)

-- 3ª definición
-- =====

solitario3 :: Ord a => [a] -> a
solitario3 [x] = x
solitario3 (x1:x2:x3:xs)
    | x1 /= x2 && x2 == x3 = x1
    | x1 == x2 && x2 /= x3 = x3
    | otherwise           = solitario3 (x2:x3:xs)
solitario3 _ = error "Imposible"

-- 4ª definición
-- =====

solitario4 :: Ord a => [a] -> a
solitario4 xs
    | y1 == y2 = last ys
    | otherwise = y1
    where (y1:y2:ys) = sort xs

-- 5ª definición
-- =====

```

```

solitario5 :: Ord a => [a] -> a
solitario5 xs | null ys    = y
               | otherwise = z
  where [y:ys,z:_] = group (sort xs)

-- 6ª definición
-- =====

solitario6 :: Ord a => [a] -> a
solitario6 xs =
  head [x | x <- nub xs
          , cuenta xs x == 1]

-- 7ª definición
-- =====

solitario7 :: Ord a => [a] -> a
solitario7 (a:b:xs)
  | a == b      = solitario7 (b:xs)
  | elem a (b:xs) = b
  | elem b (a:xs) = a
solitario7 [_ ,b] = b
solitario7 _      = error "Imposible"

-- Equivalencia
-- =====

-- Propiedad de equivalencia
prop_solitario_equiv :: Property
prop_solitario_equiv =
  forAll listaSolitaria (\xs -> solitario xs == solitario2 xs &&
                                solitario xs == solitario3 xs &&
                                solitario xs == solitario4 xs &&
                                solitario xs == solitario5 xs &&
                                solitario xs == solitario6 xs &&
                                solitario xs == solitario7 xs)

-- Generador de listas con al menos 3 elementos y todos iguales menos
-- uno. Por ejemplo,
--   λ> sample listaSolitaria

```

```

--      [1,0,0,0,0]
--      [0,0,-1,0,0,0]
--      [4,1,1,1]
--      [6,6,4,6]
--      [8,8,8,8,8,-4,8,8,8,8,8,8]
--      ...
listaSolitaria :: Gen [Int]
listaSolitaria = do
  n <- arbitrary
  m <- arbitrary 'suchThat' (\a -> n + a > 2)
  x <- arbitrary
  y <- arbitrary 'suchThat' (\a -> a /= x)
  return (replicate n x ++ [y] ++ replicate m x)

-- Comprobación:
--      λ> quickCheck prop_solitario_equiv
--      +++ OK, passed 100 tests.

-- Comparación de eficiencia:
--      λ> solitario (replicate (5*10^3) 1 ++ [2])
--      2
--      (5.47 secs, 3,202,688,152 bytes)
--      λ> solitario2 (replicate (5*10^3) 1 ++ [2])
--      2
--      (2.08 secs, 1,401,603,960 bytes)
--      λ> solitario3 (replicate (5*10^3) 1 ++ [2])
--      2
--      (0.04 secs, 3,842,240 bytes)
--      λ> solitario4 (replicate (5*10^3) 1 ++ [2])
--      2
--      (0.02 secs, 1,566,472 bytes)
--      λ> solitario5 (replicate (5*10^3) 1 ++ [2])
--      2
--      (0.01 secs, 927,064 bytes)
--      λ> solitario6 (replicate (5*10^3) 1 ++ [2])
--      2
--      (0.01 secs, 1,604,176 bytes)
--      λ> solitario7 (replicate (5*10^3) 1 ++ [2])
--      2
--      (0.01 secs, 1,923,440 bytes)

```

```
--  
-- λ> solitario3 (replicate (5*10^6) 1 ++ [2])  
-- 2  
-- (4.62 secs, 3,720,123,560 bytes)  
-- λ> solitario4 (replicate (5*10^6) 1 ++ [2])  
-- 2  
-- (1.48 secs, 1,440,124,240 bytes)  
-- λ> solitario5 (replicate (5*10^6) 1 ++ [2])  
-- 2  
-- (1.40 secs, 1,440,125,936 bytes)  
-- λ> solitario6 (replicate (5*10^6) 1 ++ [2])  
-- 2  
-- (2.65 secs, 1,480,125,032 bytes)  
-- λ> solitario7 (replicate (5*10^6) 1 ++ [2])  
-- 2  
-- (2.21 secs, 1,800,126,224 bytes)  
--  
-- λ> solitario5 (2 : replicate (5*10^6) 1)  
-- 2  
-- (1.38 secs, 1,520,127,864 bytes)  
-- λ> solitario6 (2 : replicate (5*10^6) 1)  
-- 2  
-- (1.18 secs, 560,127,664 bytes)  
-- λ> solitario7 (2 : replicate (5*10^6) 1)  
-- 2  
-- (0.29 secs, 280,126,888 bytes)
```





## Ejercicio 15

### Números colinas

#### Ejercicio propuesto el 29-11-18

Se dice que un número natural  $n$  es una colina si su primer dígito es igual a su último dígito, los primeros dígitos son estrictamente creciente hasta llegar al máximo, el máximo se puede repetir y los dígitos desde el máximo al final son estrictamente decrecientes.

Definir la función

---

```
esColina :: Integer -> Bool
```

---

tal que (esColina  $n$ ) se verifica si  $n$  es un número colina. Por ejemplo,

---

```
esColina 12377731 == True
esColina 1237731  == True
esColina 123731   == True
esColina 12377730 == False
esColina 12377730 == False
esColina 10377731 == False
esColina 12377701 == False
esColina 33333333 == True
```

---

## Soluciones

```

import Data.Char (digitToInt)
import Test.QuickCheck

-- 1ª definición
-- =====

esColina :: Integer -> Bool
esColina n =
  head ds == last ds &&
  esCreciente xs &&
  esDecreciente ys
  where ds = digitos n
        m  = maximum ds
        xs = takeWhile (<m) ds
        ys = dropWhile (==m) (dropWhile (<m) ds)

-- (digitos n) es la lista de los dígitos de n. Por ejemplo,
--   digitos 425 == [4,2,5]
digitos :: Integer -> [Int]
digitos n = map digitToInt (show n)

-- (esCreciente xs) se verifica si la lista xs es estrictamente
-- creciente. Por ejemplo,
--   esCreciente [2,4,7] == True
--   esCreciente [2,2,7] == False
--   esCreciente [2,1,7] == False
esCreciente :: [Int] -> Bool
esCreciente xs = and [x < y | (x,y) <- zip xs (tail xs)]

-- (esDecreciente xs) se verifica si la lista xs es estrictamente
-- decreciente. Por ejemplo,
--   esDecreciente [7,4,2] == True
--   esDecreciente [7,2,2] == False
--   esDecreciente [7,1,2] == False
esDecreciente :: [Int] -> Bool
esDecreciente xs = and [x > y | (x,y) <- zip xs (tail xs)]

-- 2ª definición

```

```
-- =====

esColina2 :: Integer -> Bool
esColina2 n =
  head ds == last ds &&
  null (dropWhile (==(-1)) (dropWhile (==0) (dropWhile (==1) xs)))
  where ds = digitos n
        xs = [signum (y-x) | (x,y) <- zip ds (tail ds)]

-- Equivalencia
-- =====

-- La propiedad de equivalencia es
prop_esColina :: Integer -> Property
prop_esColina n =
  n >= 0 ==> esColina n == esColina2 n

-- La comprobación es
--   λ> quickCheck prop_esColina
--   +++ OK, passed 100 tests.
```



# Ejercicio 16

## Raíz cúbica entera

### Ejercicio propuesto el 30-11-18

Un número  $x$  es un cubo si existe un  $y$  tal que  $x = y^3$ . Por ejemplo, 8 es un cubo porque  $8 = 2^3$ .

Definir la función

---

```
raizCubicaEntera :: Integer -> Maybe Integer.
```

---

tal que (raizCubicaEntera x n) es justo la raíz cúbica del número natural x, si x es un cubo y Nothing en caso contrario. Por ejemplo,

---

raizCubicaEntera 8	==	Just 2
raizCubicaEntera 9	==	Nothing
raizCubicaEntera 27	==	Just 3
raizCubicaEntera 64	==	Just 4
raizCubicaEntera (2^30)	==	Just 1024
raizCubicaEntera (10^9000)	==	Just (10^3000)
raizCubicaEntera (5 + 10^9000)	==	Nothing

---

## Soluciones

```
import Data.Numbers.Primes (primeFactors)
import Data.List           (group)
```

```
import Test.QuickCheck
```

```
-- 1ª definición
```

```
-- =====
```

```
raizCubicaEntera :: Integer -> Maybe Integer
```

```
raizCubicaEntera x = aux 0
```

```
  where aux y | y^3 > x    = Nothing
              | y^3 == x   = Just y
              | otherwise = aux (y+1)
```

```
-- 2ª definición
```

```
-- =====
```

```
raizCubicaEntera2 :: Integer -> Maybe Integer
```

```
raizCubicaEntera2 x
```

```
  | y^3 == x = Just y
  | otherwise = Nothing
  where (y:_) = dropWhile (\z -> z^3 < x) [0..]
```

```
-- 3ª definición
```

```
-- =====
```

```
raizCubicaEntera3 :: Integer -> Maybe Integer
```

```
raizCubicaEntera3 1 = Just 1
```

```
raizCubicaEntera3 x = aux (0,x)
```

```
  where aux (a,b) | d == x    = Just c
                  | c == a    = Nothing
                  | d < x     = aux (c,b)
                  | otherwise = aux (a,c)
```

```
  where c = (a+b) `div` 2
        d = c^3
```

```
-- 4ª definición
```

```
-- =====
```

```
raizCubicaEntera4 :: Integer -> Maybe Integer
```

```
raizCubicaEntera4 x
```

```
  | y^3 == x = Just y
  | otherwise = Nothing
```

```

where y = floor ((fromIntegral x)**(1 / 3))

-- Nota. La definición anterior falla para números grandes. Por ejemplo,
--   λ> raizCubicaEntera4 (2^30)
--   Nothing
--   λ> raizCubicaEntera (2^30)
--   Just 1024

-- 5ª definición
-- =====

raizCubicaEntera5 :: Integer -> Maybe Integer
raizCubicaEntera5 x
  | all (==0) [length as `mod` 3 | as <- ass] =
    Just (product [a^((1 + length as) `div` 3) | (a:as) <- ass])
  | otherwise = Nothing
where ass = group (primeFactors x)

-- Equivalencia
-- =====

-- La propiedad es
prop_raizCubicaEntera :: Integer -> Property
prop_raizCubicaEntera x =
  x >= 0 ==>
  and [raizCubicaEntera x == f x | f <- [ raizCubicaEntera2
                                           , raizCubicaEntera3]]

-- La comprobación es
--   λ> quickCheck prop_raizCubicaEntera
--   +++ OK, passed 100 tests.

-- Comparación de eficiencia
-- =====

--   λ> raizCubicaEntera (10^18)
--   Just 1000000
--   (1.80 secs, 1,496,137,192 bytes)
--   λ> raizCubicaEntera2 (10^18)
--   Just 1000000

```

```
-- (0.71 secs, 712,134,128 bytes)
-- λ> raizCubicaEntera3 (10^18)
-- Just 1000000
-- (0.01 secs, 196,424 bytes)
--
-- λ> raizCubicaEntera2 (5^27)
-- Just 1953125
-- (1.42 secs, 1,390,760,920 bytes)
-- λ> raizCubicaEntera3 (5^27)
-- Just 1953125
-- (0.00 secs, 195,888 bytes)
--
-- λ> raizCubicaEntera3 (10^9000) == Just (10^3000)
-- True
-- (2.05 secs, 420,941,368 bytes)
-- λ> raizCubicaEntera3 (5 + 10^9000) == Nothing
-- True
-- (2.08 secs, 420,957,576 bytes)
-- λ> raizCubicaEntera5 (5 + 10^9000) == Nothing
-- True
-- (0.03 secs, 141,248 bytes)
```




## Ejercicio 17

# Numeración de los árboles binarios completos

### Ejercicio propuesto el 3 de Diciembre de 2018

Un [árbol binario completo](#) es un árbol binario que tiene todos los nodos posibles hasta el penúltimo nivel, y donde los elementos del último nivel están colocados de izquierda a derecha sin dejar huecos entre ellos.

La numeración de los árboles binarios completos se realiza a partir de la raíz, recorriendo los niveles de izquierda a derecha. Por ejemplo,



---

Los árboles binarios se puede representar mediante el siguiente tipo

---

```
data Arbol = H
           | N Int Arbol Arbol
           deriving (Show, Eq)
```

---

Definir la función

---

```
arbolBinarioCompleto :: Int -> Arbol
```

---

tal que (arbolBinarioCompleto n) es el árbol binario completo con n nodos. Por ejemplo,

---

```
λ> arbolBinarioCompleto 4
N 1 (N 2 (N 4 H H) H) (N 3 H H)
λ> pPrint (arbolBinarioCompleto 9)
N 1
  (N 2
    (N 4
      (N 8 H H)
      (N 9 H H))
    (N 5 H H))
  (N 3
    (N 6 H H)
    (N 7 H H))
```

---

## Soluciones

```
data Arbol = H
           | N Int Arbol Arbol
  deriving (Eq, Show)

arbolBinarioCompleto :: Int -> Arbol
arbolBinarioCompleto n = aux 1
  where aux i | i <= n    = N i (aux (2*i)) (aux (2*i+1))
            | otherwise = H
```

## Ejercicio 18

# Posiciones en árboles binarios

### Ejercicio propuesto el 2 de diciembre de 2018


Los árboles binarios con datos en los nodos se definen por

---

```
data Arbol a = H
              | N a (Arbol a) (Arbol a)
  deriving (Eq, Show)
```

---

Por ejemplo, el árbol



se representa por

---

```
ejArbol :: Arbol Int
ejArbol = N 3
          (N 0
            (N 5
              (N 2 H H)
            )
          )
```

---

```

      (N 4 H H))
    (N 0 H H))
  (N 5
    (N 0 H H)
    (N 3 H H))

```

---

Cada posición de un elemento de un árbol es una lista de movimientos hacia la izquierda o hacia la derecha. Por ejemplo, la posición de 4 en el árbol anterior es [I,I,D].

Los tipos de los movimientos y de las posiciones se definen por

---

```

data Movimiento = I | D deriving (Show, Eq)
type Posicion   = [Movimiento]

```

---

Definir la función

---

```

posiciones :: Eq b => b -> Arbol b -> [Posicion]
\end{solucion}

```

tal que (posiciones n a) es la lista de las posiciones del elemento n en el árbol a. Por ejemplo,

```

\begin{descripcion}
posiciones 0 ejArbol == [[I],[I,D],[D,I]]
posiciones 2 ejArbol == [[I,I,I]]
posiciones 3 ejArbol == [[],[D,D]]
posiciones 4 ejArbol == [[I,I,D]]
posiciones 5 ejArbol == [[I,I],[D]]
posiciones 1 ejArbol == []

```

---

## Soluciones

```

import Data.List (nub)
import Test.QuickCheck

data Arbol a = H
             | N a (Arbol a) (Arbol a)
  deriving (Eq, Show)

```

```
ejArbol :: Arbol Int
```

```
ejArbol = N 3
          (N 0
            (N 5
              (N 2 H H)
              (N 4 H H))
            (N 0 H H))
          (N 5
            (N 0 H H)
            (N 3 H H))
```

```
data Movimiento = I | D deriving (Show, Eq, Ord)
```

```
type Posicion = [Movimiento]
```

```
-- 1ª solución
```

```
-- =====
```

```
posiciones :: Eq b => b -> Arbol b -> [Posicion]
```

```
posiciones n a = aux n a [[]]
```

```
  where aux _ H _ = []
        aux n' (N x i d) cs | x == n' = cs ++
                                [I:xs | xs <- aux n' i cs] ++
                                [D:xs | xs <- aux n' d cs]
                                | otherwise = [I:xs | xs <- aux n' i cs] ++
                                                [D:xs | xs <- aux n' d cs]
```

```
-- 2ª solución
```

```
-- =====
```

```
posiciones2 :: Eq b => b -> Arbol b -> [Posicion]
```

```
posiciones2 n a = aux n a [[]]
```

```
  where aux _ H _ = []
        aux n' (N x i d) cs | x == n' = cs ++ ps
                                | otherwise = ps
        where ps = [I:xs | xs <- aux n' i cs] ++
                    [D:xs | xs <- aux n' d cs]
```

```
-- 3ª solución
```

```
-- =====
```

```

posiciones3 :: Eq b => b -> Arbol b -> [Posicion]
posiciones3 n a = aux n a [[]]
  where aux _ H _ = []
        aux n' (N x i d) cs | x == n' = cs ++ ps
                              | otherwise = ps
        where ps = map (I:) (aux n' i cs) ++
                  map (D:) (aux n' d cs)

-- Equivalencia
-- =====

-- Generador de árboles
instance Arbitrary a => Arbitrary (Arbol a) where
  arbitrary = sized genArbol

genArbol :: (Arbitrary a, Integral a1) => a1 -> Gen (Arbol a)
genArbol 0 = return H
genArbol n | n > 0 = N <$> arbitrary <*> subarbol <*> subarbol
  where subarbol = genArbol (div n 2)
genArbol _ = error "Imposible"

-- La propiedad es
prop_posiciones_equiv :: Arbol Int -> Bool
prop_posiciones_equiv a =
  and [posiciones n a == posiciones2 n a | n <- xs] &&
  and [posiciones n a == posiciones3 n a | n <- xs]
  where xs = take 3 (elementos a)

-- (elementos a) son los elementos del árbol a. Por ejemplo,
-- elementos ejArbol == [3,0,5,2,4]
elementos :: Eq b => Arbol b -> [b]
elementos H = []
elementos (N x i d) = nub (x : elementos i ++ elementos d)

-- La comprobación es
-- λ> quickCheck prop_posiciones_equiv
-- +++ OK, passed 100 tests.

```


## Ejercicio 19

# Posiciones en árboles binarios completos

### Ejercicio propuesto el 5 de diciembre de 2018

Un [árbol binario completo](#) es un árbol binario que tiene todos los nodos posibles hasta el penúltimo nivel, y donde los elementos del último nivel están colocados de izquierda a derecha sin dejar huecos entre ellos.

La numeración de los árboles binarios completos se realiza a partir de la raíz, recorriendo los niveles de izquierda a derecha. Por ejemplo,



Los árboles binarios se puede representar mediante el siguiente tipo

---

```
data Arbol = H
           | N Integer Arbol Arbol
deriving (Show, Eq)
```

---

Cada posición de un elemento de un árbol es una lista de movimientos hacia la izquierda o hacia la derecha. Por ejemplo, la posición de 9 en el árbol anterior es [I,I,D].

Los tipos de los movimientos y de las posiciones se definen por

---

```
data Movimiento = I | D deriving (Show, Eq)
type Posicion   = [Movimiento]
```

---

Definir la función

---

```
posicionDeElemento :: Integer -> Posicion
```

---

tal que (posicionDeElemento n) es la posición del elemento n en el árbol binario completo. Por ejemplo,

---

```
posicionDeElemento 6 == [D,I]
posicionDeElemento 7 == [D,D]
posicionDeElemento 9 == [I,I,D]
posicionDeElemento 1 == []

length (posicionDeElemento (10^50000)) == 166096
```

---

## Soluciones

```
import Test.QuickCheck

data Arbol = H
           | N Integer Arbol Arbol
  deriving (Eq, Show)

data Movimiento = I | D deriving (Show, Eq)

type Posicion = [Movimiento]

-- 1ª solución
-- =====

posicionDeElemento :: Integer -> Posicion
```



```

posicionDeElemento n =
  head (posiciones n (arbolBinarioCompleto n))

-- (arbolBinarioCompleto n) es el árbol binario completo con n
-- nodos. Por ejemplo,
--   λ> arbolBinarioCompleto 4
--   N 1 (N 2 (N 4 H H) H) (N 3 H H)
--   λ> pPrint (arbolBinarioCompleto 9)
--   N 1
--     (N 2
--       (N 4
--         (N 8 H H)
--         (N 9 H H))
--       (N 5 H H))
--     (N 3
--       (N 6 H H)
--       (N 7 H H))
arbolBinarioCompleto :: Integer -> Arbol
arbolBinarioCompleto n = aux 1
  where aux i | i <= n    = N i (aux (2*i)) (aux (2*i+1))
            | otherwise = H

-- (posiciones n a) es la lista de las posiciones del elemento n
-- en el árbol a. Por ejemplo,
--   posiciones 9 (arbolBinarioCompleto 9) == [[I,I,D]]
posiciones :: Integer -> Arbol -> [Posicion]
posiciones n a = aux n a [[]]
  where aux _ H _ = []
        aux n' (N x i d) cs | x == n' = cs ++ ps
                           | otherwise = ps
        where ps = map (I:) (aux n' i cs) ++
                  map (D:) (aux n' d cs)

-- 2ª solución
-- =====

posicionDeElemento2 :: Integer -> Posicion
posicionDeElemento2 1 = []
posicionDeElemento2 n
  | even n    = posicionDeElemento2 (n `div` 2) ++ [I]

```

```

    | otherwise = posicionDeElemento2 (n `div` 2) ++ [D]

-- 3ª solución
-- =====

posicionDeElemento3 :: Integer -> Posicion
posicionDeElemento3 = reverse . aux
  where aux 1 = []
        aux n | even n    = I : aux (n `div` 2)
              | otherwise = D : aux (n `div` 2)

-- 4ª solución
-- =====

posicionDeElemento4 :: Integer -> Posicion
posicionDeElemento4 n =
  [f x | x <- tail (reverse (binario n))]
  where f 0 = I
        f 1 = D
        f _ = error "Imposible"

-- (binario n) es la lista de los dígitos de la representación binaria
-- de n. Por ejemplo,
--   binario 11 == [1,1,0,1]
binario :: Integer -> [Integer]
binario n
  | n < 2    = [n]
  | otherwise = n `mod` 2 : binario (n `div` 2)

-- Equivalencia
-- =====

-- La propiedad es
prop_posicionDeElemento_equiv :: Positive Integer -> Bool
prop_posicionDeElemento_equiv (Positive n) =
  posicionDeElemento n == posicionDeElemento2 n &&
  posicionDeElemento n == posicionDeElemento3 n &&
  posicionDeElemento n == posicionDeElemento4 n

-- La comprobación es

```

```
--      λ> quickCheck prop_posicionDeElemento_equiv
--      +++ OK, passed 100 tests.

-- Comparación de eficiencia
-- =====

--      λ> posicionDeElemento (10^7)
--      [I,I,D,D,I,I,I,D,I,I,D,I,D,D,I,D,I,I,I,I,I,I,I]
--      (5.72 secs, 3,274,535,328 bytes)
--      λ> posicionDeElemento2 (10^7)
--      [I,I,D,D,I,I,I,D,I,I,D,I,D,D,I,D,I,I,I,I,I,I,I]
--      (0.01 secs, 189,560 bytes)
--      λ> posicionDeElemento3 (10^7)
--      [I,I,D,D,I,I,I,D,I,I,D,I,D,D,I,D,I,I,I,I,I,I,I]
--      (0.01 secs, 180,728 bytes)
--      λ> posicionDeElemento4 (10^7)
--      [I,I,D,D,I,I,I,D,I,I,D,I,D,D,I,D,I,I,I,I,I,I,I]
--      (0.01 secs, 184,224 bytes)
--
--      λ> length (posicionDeElemento2 (10^4000))
--      13287
--      (2.80 secs, 7,672,011,280 bytes)
--      λ> length (posicionDeElemento3 (10^4000))
--      13287
--      (0.03 secs, 19,828,744 bytes)
--      λ> length (posicionDeElemento4 (10^4000))
--      13287
--      (0.03 secs, 18,231,536 bytes)
--
--      λ> length (posicionDeElemento3 (10^50000))
--      166096
--      (1.34 secs, 1,832,738,136 bytes)
--      λ> length (posicionDeElemento4 (10^50000))
--      166096
--      (1.70 secs, 1,812,806,080 bytes)
```